```
PROGRAM

! Initiated by Lingli Huang on 08/18/2014;

! Checked by Zhoumeng Lin on 10/21/2014;


INITIAL

! code that is executed once at the beginning of a simulation run goes here

!! Physiological parameters

! Blood flow rates

 CONSTANT QCC = 5.0 ! Cardiac output (L/h/kg) (Average of Buur et al., 2005 and Upton, 2008)

 CONSTANT QLC = 0.27 ! Fraction of blood flow to the liver (Average of Buur et al., 2005 and Upton, 2008)

 CONSTANT QKC = 0.12 ! Fraction of blood flow to the kidneys (Average of Buur et al., 2005 and Upton, 2008)

 CONSTANT QFC = 0.13 ! Fraction of blood flow to the fat (Average of Buur et al., 2005 and Upton, 2008)

 CONSTANT QMC = 0.25 ! Fraction of blood flow to the muscle (Average of Buur et al., 2005 and Upton, 2008)

 CONSTANT QGC = 0.18 ! Fraction of blood flow to the GI tract (Tranquillity et al, 1982)


! Tissue volumes

 CONSTANT BW = 50 ! Body weight (kg) (Li et al., 2013)

 CONSTANT VLC = 0.025 ! Fractional liver tissue (Average of Buur et al., 2005 and Upton, 2008)

 CONSTANT VKC = 0.004 ! Fractional kidney tissue (Average of Buur et al., 2005 and Upton, 2008)

 CONSTANT VFC = 0.32 ! Fractional fat tissue (Average of Buur et al., 2005 and Upton, 2008)

 CONSTANT VMC = 0.40 ! Fractional muscle tissue (Average of Buur et al., 2005 and Upton, 2008)

 CONSTANT VGC = 0.05 ! Fractional GI tract (Upton,2008)
```

CONSTANT VbloodC = 0.06 ! Blood volume, fraction of BW(Average of Buur et al., 2005 and Upton, 2008)

! Mass Transfer Parameters (Chemical-specific parameters)

CONSTANT MW = 271 ! (g/mol,Molecular weight)

! Partition coefficients for CYA (PC, tissue:plasma)

CONSTANT PLcya = 1.0 ! Liver:plasma PC (Estimated based on Li et al,2013)

CONSTANT PKcya = 1.28 ! Kidney:plasma PC (Estimated based on Li et al,2013)

CONSTANT PMcya = 4.55 ! Muscle:plasma PC (Estimated based on Li et al,2013)

CONSTANT PFcya = 1.98 ! Fat:plasma PC (Estimated based on Li et al,2013)

CONSTANT POTcya = 1.0 ! GI:plasma PC (Assumed as the same with liver)

! Partition coefficients for BDCYA (PC, tissue:plasma)

CONSTANT PLbdcya = 0.91 ! Liver:plasma PC (Measured from the experiment)

CONSTANT PKbdcya = 2.87 ! Kidney:plasma PC (Measured from the experiment)

CONSTANT PMbdcya = 0.32 ! Muscle:plasma PC (Measured from the experiment)

CONSTANT PFbdcya = 0.30 ! Fat:plasma PC (Measured from the experiment)

CONSTANT POTbdcya = 0.91 ! GI:plasma PC (Assumed as the same with liver)

! Kinetic constants

! Oral absorption rate constants

CONSTANT Ka = 0.0015 ! /h, intestinal absorption rate constant for CYA

CONSTANT Ka1 = 0.0008 ! /h, intestinal absorption rate constant for BDCYA

CONSTANT Kst= 0.2 ! /h, gastric emptying rate constant

CONSTANT Kef = 0.4 ! /h, intestinal transit rate constant for CYA

CONSTANT Kef1 = 0.2 ! /h, intestinal transit rate constant for BDCYA

CONSTANT Bioavail= 0.44 ! /%, Bioavailability of CYA in pigs. (Measured by Xu et al,2012)

CONSTANT PB= 0.153 ! /, plasma protein combination ratio of CYA. (Measured by experiment)

CONSTANT PB1= 0.722 ! /, plasma protein combination ratio OF BDCYA. (Measured by experiment)

!Metabolic constants for CYA_BDCYA

CONSTANT Km1C = 0.01 ! (Liver metabolism rate constant of CYA)

CONSTANT FL = 0.17 ! (Fraction of CYA to be metabolized into BDCYA in liver)

CONSTANT Km2C = 0.04 ! (Intestinal metabolism rate constant of CYA)

CONSTANT FI = 0.2 ! (Fraction of CYA to be metabolized into BDCYA in intestine)

! Elimination rate constant

CONSTANT Clcya = 0.035 ! L/h/kg (Renal clearance rate of CYA, Measured by experiment)

CONSTANT Clbdcya = 0.0044 ! L/h/kg (Renal clearance rate of BDCYA,Measured by experiment)

CONSTANT KbC = 0.01 ! (bile elimination rate constant for CYA)

CONSTANT KbC1 = 0.01 ! (bile elimination rate constant for BDCYA)

!Parameters for various exposure scenarios

Constant PDOSEoral = 20 ! oral dose of CYA in pigs(mg/kg b.w.)

! Determine withdrawal

CONSTANT TL= 8000 ! (Tolerance of BDCYA in liver)

CONSTANT TK= 3000 ! (Tolerance of BDCYA in kidney)

CONSTANT TM= 500 ! (Tolerance of BDCYA in muscle)

```
CONSTANT TF= 2000   !  (Tolerance of BDCYA in fat)

Twithdrawal=0


END ! INITIAL

DYNAMIC

ALGORITHM IALG = 2
NSTEPS            NSTP = 10
MAXTERVAL MAXT = 1.0e9
MINTERVAL MINT = 1.0e-9
CINTERVAL CINT = 0.1

DERIVATIVE
! code for calculating the derivative goes here
! Cardiac output and blood flows to tissues (L/h)
 QC = QCC*BW ! Cardiac output
 QL = QLC*QC ! Blood flow to the liver
 QK = QKC*QC ! Blood flow to the kidney
 QF = QFC*QC ! Blood flow to the fat
 QM = QMC*QC ! Blood flow to the muscle
 QG = QGC*QC ! Blood flow to the GI tract
 QOT = QC-QL-QK-QM-QF ! Blood flow to the other tissues
```

```
! Tissue volumes (L)

 VL = VLC*BW ! Liver

 VK = VKC*BW ! Kidney

 VF = VFC*BW ! Fat

 VM = VMC*BW ! Muscle

 VG = VGC*BW ! GI tract

 Vblood = VbloodC*BW ! Blood

 VOT = BW-VL-VK-VM-VF-Vblood ! Other tissues


!Unit conversion from mg CYA to umol CYA :1mg = 3.69umol

!1mg=1000ug

!1000ug/271g/mol=3.69umol

!1mg=3.69umol


CONSTANT MWCYA = 3.69 !Molecular weight of CYA (umol/mg)




! Dosing

 DOSEoral = PDOSEoral*BW*MWCYA*Bioavail ! (umol)


!...Dosing, multiple oral gavage

 CONSTANT tlen = 0.001 ! Length of oral gavage exposure (h/day)
```

```
CONSTANT tinterval = 12 ! Varied dependent on the exposure paradigm(h)

CONSTANT Dstart = 0.0 ! Initiation day of oral gavage (day)

CONSTANT Dstop = 5 ! Termination day of oral gavage (day)

CONSTANT MAXT = 1.0 ! maximum comm. interval

CONSTANT CINTC = 0.1 ! Communication interval

CINT = CINTC    ! Communication interval


Tsim = TSTOP ! Tstop in hours

DS = Dstart*24 ! Initiation time point of oral gavage (h)

Doff = (Dstop - Dstart)*24 ! Oral gavage duration (h)


TimeOn = Dstart*24! Initiation time point of oral gavage (h)

TimeOff = Dstop*24+tlen ! Termination time point of oral gavage (h)


Exposure = PULSE(0,tinterval,tlen)*PULSE(DS, Tsim, Doff)

Rdoseoral = (Doseoral/tlen)*Exposure   !Rate of oral dosage

RAST=Rdoseoral-Kst*AST

AST=INTEG(rast,0)


RI = Kst*AST-Ka*AI-Kef*AI-Km2*AI         ! Rate of CYA in intestine

AI = Integ(RI,0) ! Amount of doseage in intestine

RAO=Ka*AI       ! Rate of absorbed by intestine

AAO = Integ(RAO,0)       ! Amount of absorbed by intestine

AEF=Kef*AI ! Amount of CYA excreted from intestine

AEF1=Kef1*AI  ! Amount of BDCYA excreted from intestine
```

```
! Elimination rate constant
 Keu = CLcya*BW
 Keu1 = Clbdcya*BW
 Km1 = Km1C*BW
 Km2 = Km2C*BW
 Kb = KbC*BW
 Kb1 = KbC1*BW

! CYA in blood compartment
 CV = (QL*CVL+QK*CVK+QM*CVM+QF*CVF+QOT*CVOT)/QC
 RA = QC*(CV-CA)
 AA = Integ(RA,0)
 CA = AA/Vblood*(1-PB)
 AUCCV = Integ(CV,0.0)

! CYA in liver compartment
 RL = QL*(CA-CVL)+RAO-Rmet-Rbile
 AL = Integ(RL,0)
 CL = AL/VL
 CVL = AL/(VL*PLcya)
 AUCCL = Integ(CL,0.0)
 CLmass=CL*MW

! Metabolism of CYA in liver
```

```
Rmet = Km1*CVL !rate of metabolism of CYA to all metabolites

Amet = integ(Rmet,0)

RCYA_BDCYA = FL*Km1*CVL !rate of metabolism of CYA to BDCYA

ACYA_BDCYA = integ(RCYA_BDCYA,0)


!Bile Excretion of CYA

Rbile = Kb*CVL !rate of bile excretion of CYA

Abile = integ(Rbile,0)


! CYA in kidney compartment

RK = QK*(CA-CVK)-Rurine

AK = Integ(RK,0)

CK = AK/VK

CVK = AK/(VK*PKcya)

AUCCK = Integ(CK,0.0)

CKmass=CK*MW


! Urinary excretion of CYA

Rurine = Keu*CVK

Aurine = Integ(Rurine,0)


! CYA in muscle compartment

RM = QM*(CA-CVM)

AM = Integ(RM,0)

CM = AM/(VM)
```

```
CVM = AM/(VM*PMcya)

AUCCM = Integ(CM,0.0)

CMmass=CM*MW


! CYA in fat compartment

RF = QF*(CA-CVF)

AF = Integ(RF,0)

CF = AF/(VF)

CVF = AF/(VF*PFcya)

AUCCF = Integ(CF,0.0)

CFmass=CF*MW


! CYA in other tissue compartment

ROT = QOT*(CA-CVOT)

AOT = Integ(ROT,0)

COT = AOT/VOT

CVOT = AOT/(VOT*POTcya)

COTmass=COT*MW


! Mass balance

Qbal = QC-QL-QK-QOT-QM-QF

Tmass = AA+AL+AK+AM+AF+AOT+Aurine+Amet+Abile

BALCYA = AAO-Tmass ! Flow-limited model mass balance
```

```
!************Submodel for BDCYA**************************************

! BDCYA in blood

! BDCYA in blood compartment

 CV1 = (QL*CVL1+QK*CVK1+QM*CVM1+QF*CVF1+QOT*CVOT1)/QC

 RA1 = QC*(CV1-CA1)

 AA1 = Integ(RA1,0)

 CA1 = AA1/Vblood*(1-PB1)

 AUCCV1 = Integ(CV1,0.0)


! BDCYA in liver compartment

! AL1 =amount of BDCYA in the liver

 RL1 = QL*(CA1-CVL1)+Rmet*FL+RAO1-Rbile1

 AL1 = Integ(RL1,0)

 CL1 = AL1/VL

 CVL1 = AL1/(VL*PLbdcya)

 AUCCL1 = Integ(CL1,0.0)

 CLmass1=CL1*MW


! AAO1 =Amount of BDCYA absorbed from the intestine

 RI1 = Km2*FI*AI-Ka1*AI1-kef1*AI1 ! Rate of BDCYA in intestine

 AI1 = Integ(RI1,0) ! Amount of BDCYA in intestine

 RAO1 = Ka1*AI1 ! Rate of BDCYA absorbed by intestine

 AAO1 = Integ(RAO1,0) ! Amount of BDCYA absorbed by intestine
```

```
!Bile Excretion of BDCYA

 Rbile1 = Kb1*CVL1 !rate of bile excretion of BDCYA

 Abile1 = integ(Rbile1,0)


! BDCYA in kidney compartment

 RK1 = QK*(CA1-CVK1)-Rurine1

 AK1 = Integ(RK1,0)

 CK1 = AK1/VK

 CVK1 = AK1/(VK*PKbdcya)

 AUCCK1 = Integ(CK1,0.0)

 CKmass1=CK1*MW


! Urinary excretion of BDCYA

 Rurine1 = Keu1*CVK1

 Aurine1 = Integ(Rurine1,0)


! CYA in muscle compartment

 RM1 = QM*(CA1-CVM1)

 AM1 = Integ(RM1,0)

 CM1 = AM1/(VM)

 CVM1 = AM1/(VM*PMbdcya)

 AUCCM1 = Integ(CM1,0.0)

 CMmass1=CM1*MW


! CYA in fat compartment
```

```
RF1 = QF*(CA1-CVF1)

AF1 = Integ(RF1,0)

CF1 = AF1/(VF)

CVF1 = AF1/(VF*PFbdcya)

AUCCF1 = Integ(CF1,0.0)

CFmass1=CF1*MW


! CYA in other tissue compartment

ROT1 = QOT*(CA1-CVOT1)

AOT1 = Integ(ROT1,0)

COT1 = AOT1/(VOT)

CVOT1 = AOT1/(VOT*POTbdcya)

COTmass1=COT1*MW


! Mass balance of BDCYA

! Tmass1=mass balance of BDCYA


Tmass1 = AA1+AL1+AK1+AM1+AF1+AOT1+Aurine1+Abile1

BALBDCYA = ACYA_BDCYA+AAO1-Tmass1 ! Flow-limited model mass balance


END ! DERIVATIVE


! Add discrete events here as needed
```

```
!DISCRETE

!END


! code that is executed once at each communication interval goes here


CONSTANT TSTOP = 300.0

TERMT (T .GE. TSTOP, 'checked on communication interval: REACHED TSTOP')


END ! DYNAMIC


TERMINAL

! code that is executed once at the end of a simulation run goes here

END ! TERMINAL


END ! PROGRAM
```